エクゾチックオプション価格のシミュレーション　その４ －パワーオプションの価格式－

Numerical Tests on the Exotic Option Pricing Models － 4 －
The Power Option Pricing Model

慶応義塾大学理工学部
竹内寿一郎

１．はじめに

前回はスプレッドオプションということで、挙動が幾何ブラウン運動ではない、やや異なったテーマをとりあげたが、今回は昨年度のシンポジウムで発表したプレインバニラオプションを拡張した、行使価格を K^N としたパワーオプションについて述べ、$N=2$ のときのこれがプレインバニラオプション [1] の積分で表わせることも合わせて報告する。

２．パワーオプションの価格式

オプション満期 T におけるオプション価格を K としたとき、ペイオフを $Max\{S(T)^N - K^N, 0\}$ の期待値とするのがパワーオプションである。

$S(T)$ が幾何ブラウン運動に従い、r をリスクフリーの金利としたとき、その変動が次の式で表わせるならば、

(1)　$$dS = rSdt + \sigma Sd\hat{z}$$

の仮定のもとで、

(2)　$$\frac{\ln\frac{S(T)}{S(t)} - (r - \frac{\sigma^2}{2})(T-t)}{\sigma\sqrt{T-t}} \quad \sim \quad N(0, 1^2)$$

に従うことがわかる。この関係を用いてペイオフの期待値を計算する。
オプション満期 T におけるパワーオプションの価格 $S(T)$ の確率密度関数を $f(S(T))$ とすると、ペイオフは $Max\{S(T)^N - K^N, 0\}$ の期待値となるから、コールオプションの価格 C^* は、

(3)　$$\begin{aligned} C^* &= \int_{-\infty}^{\infty} Max\{S(T)^N - K^N, 0\} f(S(T)) dS(T) \\ &= \int_{K}^{\infty} \{S(T)^N - K^N\} f(S(T)) dS(T) \end{aligned}$$

(3) 式の 2 行目は Max をはずす為に、積分範囲を ($-\infty$ から ∞) を (K から ∞) に変えたものである。ここで (2) から求めた次式の $S(T)$、および x_0、を用いて

(4)　$$\left\{ \begin{array}{lcl} S(T) &=& S(t)e^{x\sigma\sqrt{T-t}+(r-\sigma^2/2)(T-t)} \\ x_0 &=& \dfrac{\ln(K/S(t)) - (r-\sigma^2/2)(T-t)}{\sigma\sqrt{T-t}} \end{array} \right.$$

により (3) を書き直すと、

(5)　$$\begin{aligned} C^* &= \int_{x_0}^{\infty} \{\left(S(t)e^{x\sigma\sqrt{T-t}+(r-\sigma^2/2)(T-t)}\right)^N - K^N\}\phi(x)dx \\ &= \int_{x_0}^{\infty} \left(S(t)^N e^{Nx\sigma\sqrt{T-t}+(r-\sigma^2/2)(T-t)} - K^N\right)\phi(x)dx \\ &= S(t)^N \int_{x_0}^{\infty} e^{Nx\sigma\sqrt{T-t}+(r-\sigma^2/2)(T-t)}\phi(x)dx - K^N \int_{x_0}^{\infty}\phi(x)dx \end{aligned}$$

ここで、$\phi(x)$ は標準正規分布の密度関数である。
この式の第１項の被積分関数を密度関数の完全平方の形に直すと、

(6) $$e^{Nx\sigma\sqrt{T-t}+(r-\sigma^2/2)(T-t)}\phi(x)=e^{N\left\{r+\frac{1}{2}(N-1)\sigma^2\right\}(T-t)}\phi(x-N\sigma\sqrt{T-t})$$

であるから、(5) 式は、

(7) $$\begin{aligned}C^*&=S(t)^N\int_{x_0}^{\infty}e^{N\left\{r+\frac{1}{2}(N-1)\sigma^2\right\}(T-t)}\phi(x-N\sigma\sqrt{T-t})dx-K^N\int_{x_0}^{\infty}\phi(x)dx\\&=S(t)^Ne^{N\left\{r+\frac{1}{2}(N-1)\sigma^2\right\}(T-t)}\int_{x_0-N\sigma\sqrt{T-t}}^{\infty}\phi(x)dx-K^N\int_{x_0}^{\infty}\phi(x)dx\\&=S(t)^Ne^{N\left\{r+\frac{1}{2}(N-1)\sigma^2\right\}(T-t)}\Phi(-x_0+N\sigma\sqrt{T-t})-K^N\Phi(-x_0)\end{aligned}$$

したがって、この現在価値をとることによりコールオプション価格 C を得る。

(8) $$\begin{aligned}C=C^*e^{-r(T-t)}&=S(t)^Ne^{\left\{(N-1)r+\frac{1}{2}N(N-1)\sigma^2\right\}(T-t)}\Phi(-x_0+N\sigma\sqrt{T-t})-K^N\Phi(-x_0)\\&=S(t)^Ne^{\left\{(N-1)r+\frac{1}{2}N(N-1)\sigma^2\right\}(T-t)}\Phi(d+N\sigma\sqrt{T-t})-K^N\Phi(d)\end{aligned}$$

ここで d は、

(9) $$d=-x_0=\frac{\ln\frac{S(t)}{K}+(r-\frac{\sigma^2}{2})(T-t)}{\sigma\sqrt{T-t}}$$

である。

３．$N=2$ のときのパワーとプレインバニラのオプションの価格式との関係

$G(K)$ を $N=2$ のときのパワーオプション価格式、$H(K)$ をプレインバニラオプションの価格式とし、双方ともに行使価格 K の関数とみなすことにする。

(10) $$\begin{cases}G(K)\ :\ Max\{S(T)^2-K^2,0\}\ \text{の期待値でパワーオプションの価格}\\H(K)\ :\ Max\{S(T)-K,0\}\ \text{の期待値でプレインバニラオプションの価格}\end{cases}$$

(11) $$G(K)=e^{-r(T-t)}\int_K^{\infty}\{S(T)^2-K^2\}f(S(T))dS(T)$$

を K で微分すると、

(12) $$\begin{aligned}G'(K)&=e^{-r(T-t)}\left\{-K^2f(K)-2K\int_K^{\infty}f(S(T))dS(T)+K^2f(K)\right\}\\&=-2e^{-r(T-t)}K\int_K^{\infty}f(S(T))dS(T)\end{aligned}$$

一方、プレインバニラオプションでは、

(13) $$H(K)=e^{-r(T-t)}\int_K^{\infty}\{S(T)-K\}f(S(T))dS(T)$$

を K で微分すると、

(14) $$\begin{aligned}H'(K)&=e^{-r(T-t)}\left\{-Kf(K)-\int_K^{\infty}f(S(T))dS(T)+Kf(K)\right\}\\&=-e^{-r(T-t)}\int_K^{\infty}f(S(T))dS(T)\end{aligned}$$

この関係から次式を得る。

(15) $$G'(K)=2K\cdot H'(K)$$

そこでこれを積分することにより、次の関係式を求めることが出来る。

(16) $$\begin{aligned}G(K)&=-\int_K^{\infty}G'(x)dx=-\int_K^{\infty}2x\cdot H'(x)dx\\&=-\left\{[2x\cdot H(x)]_K^{\infty}-\int_K^{\infty}2H(x)dx\right\}\\&=2K\cdot H(K)+2\int_K^{\infty}H(x)dx\end{aligned}$$

すなわち、$N=2$ のパワーオプションの価格は、プレインバニラオプション価格に加えて、それら行使価格 K 以上のを加算したものから得られることを示している。

４．パワーオプションのプライシングの J プログラム

2006 年 12 月のシンポジウム [2] では J503 の関数リストを掲載したが、ここでは J601 バージョンで関数を作成した。大きな違いはこれまでの J では左右の引数を x. y. を用いていたが、601 以降はピリオドを使わず、左右の引数は x y を用いるようにした。また、J の関数における引数表現を少し変更してみた。

●まず、前回の J プログラムの J601 版である。

```
NB. ==========================================
NB. Normal Distribution : High Precision
NB. Ndist(u)=NP(u>0)=Phi(u)=1-NQ(u>0)
NB. ==========================================
stnormal=:(%:@o.@2:)(%~)^@-@-:@*:
NB. When u>0 is small
NP=:3 : 0
(stnormal y)*y%(-‘%‘+‘%)/,(>:+:k),.(*:y)*>:k=.i.28
)
NB. When u>0 is large
NQ=:3 : 0
(stnormal y)*%‘+/1,,y ,.>:i.28
)
NB. Standard Normal Distribution Function Phi(u)
Ndist=:3 : 0
if. 3.3<z=:|y do. q=:NQ z else. q=:0.5-NP z end.
if. 0<:y do. q=:1-q end.
)
NB. ==========================================
NB. Yamanouti’s Formula : Percent Point u(alpha)
NB. ==========================================
Ninv_y=:3 : 0
z=.-^.4*y*(1-y)
x=.%:z*(2.0611786-5.7262204%(z+11.640595))
if. y>0.5 do. x=.-x end.
)
NB. ==========================================
NB. Cash-Digital Option Model
NB. M.Shimura & J.Takeuchi Dec. 2006
NB. ussage: cd data
NB. data is list ( 6 block)
NB. AssetPrice ExetPrice AcceptPrice Term(Month) Volatility FreeRate
NB. ex. data=. 14500 14000 15000 2 38 6
NB. ===x=====================================
```

```
CD =: 3 : 0
'a b1 b2 c d e' =: y
t=. c % 12
u=. (^. a % b1) + t* (e1=.e % 100) - -:(vol=.d % 100) ^2
p2=. u % (vol * %: t)
n2=. Ndist p2
cd=. b2 *n2 *( ^ (-e1) * t)
)
NB. =============================================
NB. Plain-Vanilla Option Model(Black Scholes)
NB. J.Takeuchi & M.Shimura Nov. 2000
NB. ussage: PV data
NB. data is list ( 5 elements)
NB. AssetPrice ExecPrice Term(Month) Volatility FreeRate
NB. ex. data=.x4500 14000 2 38 6
NB. =============================================
PV =: 3 : 0
'S K T V R'=. y
t=. T % 12
u=. (^. S % K) + t* (r=.R % 100) - -:(vol=.V % 100) ^2
d=. u % (vol * %: t)
d2=. d + vol * %: t
p1=. Ndist d2
p2=. Ndist d
C=.(S * p1 ) - K *p2 *( ^ (-r) * t)
)
NB. =============================================
NB. Normal Random Numbers
NB. Rndm_Norm Size Mu Sigma
NB. =============================================
Rndm_Norm=:3 : 0
'Num Mean Sigma'=.y
Mean+Sigma*Ninv_y"0 (?Num#10000000)%10000000
)
NB. =============================================
NB. N Cash_Digital S K A T(Month) Volatility r
NB. =============================================
Cash_Digital=:4 : 0
'S0 K A T Vol r'=.y
S=.S0[i=.0[MM=.x
```

```
label_L1.
if. T<i=.>:i do. goto_owari. end.
z=.Rndm_Norm MM,0,1
S=.S+(S*(r%100)*(1%12))+S*(Vol%100)*(%:1%12)*z
goto_L1.
label_owari.
w=.S-K
C=.(^-(r%100)*(T%12))*MM%~+/(0<w)*A
)
NB. ============================================
NB. N Plain_Vanilla S K T(Month) Volatility r
NB. ============================================
Plain_Vanilla=:4 : 0
'S0 K T Vol r'=:y
S=.S0[i=.0[MM=.x
label_L1.
if. T<i=.>:i do. goto_owari. end.
z=.Rndm_Norm MM,0,1
S=.S+(S*(r%100)*(1%12))+S*(Vol%100)*(%:1%12)*z
goto_L1.
label_owari.
w=.S-K
C=.(^-(r%100)*(T%12))*MM%~+/(0<w)#w
)
```

●使用例

```
   1000 Cash_Digital 100 90 110 12 10 10
97.2429
   CD 100 90 110 6 10 10
103.032
   1000 Cash_Digital 100 90 110 6 10 10
103.275
   1000 Plain_Vanilla 100 90 12 10 10
18.1091
   PV 100 90 12 10 10
18.6309
```

●以下は今回のパワーオプションのプライシング関数である。
シミュレーションでは、

```
NB. ============================================
NB. N Powe_Option S K n T(Month) Volatility r
```

```
NB. ==========================================
Power_Option=:4 : 0
'S0 K n T Vol r'=:y
S=.S0[i=.0[MM=.x
label_L1.
if. T<i=.>:i do. goto_owari. end.
z=.Rndm_Norm MM,0,1
S=.S+(S*(r%100)*(1%12))+S*(Vol%100)*(%:1%12)*z
goto_L1.
label_owari.
w=.S-K
C=.(^-(r%100)*(T%12))*MM%~+/(0<w)#(S^n)-K^n
)
```

正規乱数関数、Rndm_Norm を使うことに注意すること。

計算式では、

```
NB. ==========================================
NB. Power_Option Model
NB. J.Takeuchi May 2007
NB. ussage: power data
NB. data is list ( 6 elements)
NB. AssetPrice ExecPrice Power Term(Month) Volatility FreeRate
NB. ex. data=. 140 150 2 12 38 6
NB. ==========================================
Power =: 3 : 0
'S K n T V R'=. y
t=.T%12
u=.(^.S%K)+t*(r=.R%100)--:(vol=.V%100)^2
d=. u%(vol*%:t)
d2=.d+n*vol*%:t
p1=.Ndist d2
p2=.Ndist d
C=.(p1*(S^n)*^(t*(n-1)*(r+-:n**:vol))) - (K^n)*p2*(^(-r)*t)
)
```

である。

●使用例

```
   1000 Power_Option 140 150 2 12 38 6
8609.37
   1000 Power_Option 140 150 2 12 38 6
8291.67
```

```
   1000 Power_Option 140 150 2 12 38 6
7725.43
   1000 Power_Option 140 150 2 12 38 6
8356.04
   1000 Power_Option 140 150 2 12 38 6
8151.21
真の値は、
   Power 140 150 2 12 38 6
8211.57
```

かなりバラツキがやや大きいように見える。これは価格を 2 乗している影響があるためで、N 乗しなければ気にするような大きさにはならない。

【参考文献】

【1】竹内寿一郎・本田皓士（2006)：エキゾチックオプション価格のシミュレーション　その１　ーキャッシュデジタルとプレインバニラオプションー　JAPLA 2006 シンポジウム　2006.12.9 資料

【2】竹内寿一郎（2007)：エキゾチックオプション価格のシミュレーション　その２　ーダイアログボックスで決めるオプション価格の J 関数ー　JAPLA 研究会 2007.1.27 資料

【3】竹内寿一郎（2007)：エキゾチックオプション価格のシミュレーション　その３　ースプレッドオプションの価格式ー　JAPLA 研究会 2007.4.28 資料

【4】山下司 (2001)：オプションプライシングの数理　ー基礎理論と専門書のブリッジテキストー　金融財政事情研究会